

APPENDIX 2. WinBUGS (Bayesian inference Using Gibbs Sampling) language descriptions of the hierarchical Bayesian models .

Appendix 2 A.

BUGS language description of the hierarchical Bayes (HB₃) model for estimating trends and annual indices from the North American Breeding Bird Survey in Canada

{ # start model

counts, overdispersion effects, and goodness of fit stats

for(k in 1 : ncounts) {

 log(lambda[k]) <- beta[strat[k]] * (year[k] - fixedyear) + obs[strat[k],obser[k]] + eta*firstyr[k] + strata[strat[k]] + noise[k] + yeareffect[year[k],strat[k]]

 noise[k] ~ dnorm(0.0, taunoise)

 count[k] ~ dpois(lambda[k])

 fcount[k] ~ dpois(lambda[k])

 err[k] <- pow(count[k]-lambda[k],2)/lambda[k]

 ferr[k] <- pow(fcount[k]-lambda[k],2)/lambda[k]

 fzero[k] <- equals(fcount[k],0)

} # end of k-counts loop

 nfzero <- sum(fzero[1:ncounts])

 gof <- sum(err[1:ncounts])

 fgof <- sum(ferr[1:ncounts])

 diffgof <- gof-fgof

 posdiff <- step(diffgof)

 maxf <- ranked(fcount[1:ncounts],ncounts)

 meanf <- mean(fcount[1:ncounts])

 taunoise ~ dgamma(0.001,0.001)

 sdnoise <- 1 / pow(taunoise, 0.5)

observer effect hyperparameters

 mulogtauobs ~ dnorm(0.0,1.0E-6)

 taulogtauobs ~ dgamma(0.001,0.001)

first-year observer effect

 eta ~ dnorm(0.0,1.0E-6)

```

##### stratum-level effects #####

for( s in 1 : nstrata ) {

    beta[s] ~ dnorm( 0.0,1.0E-6) # long-term trend in stratum
    expbeta[s] <- exp(beta[s])
    strata[s] ~ dnorm( 0.0,1.0E-6) # stratum average abundance
    expstrata[s] <- exp(strata[s])
    overdisp[s] <- 1 + 1/(expstrata[s]*taunoise)
    tauyear[s] ~ dgamma(0.001,0.001) # precision of year-effects in stratum
    sdyear[s] <- 1 / pow(tauyear[s],0.5)

##### observer effects #####

for( i in 1 : nobservers[s] ) {
    obs[s,i] ~ dnorm( 0.0,tauobs[s])
} # end i-observer loop

    log(tauobs[s]) <- logtauobs[s]
    logtauobs[s] ~ dnorm(mulogtauobs,taulogtauobs)
    sdobs[s] <- 1 / pow(tauobs[s], 0.5)

##### year effects #####

for( y in ymin : ymax ) {
    yeaffect[y,s] ~ dnorm( 0.0, tauyear[s])
} # end y-year loop

} # end of s-strata loop

##### summary statistics - stratum level annual indices = n #####
##### n[,] values are combined across strata, weighted by stratum-area outside of BUGS #####
##### after saving the full posterior distributions of n[,] #####
for( i in 1 : nstrata ) {
    for( t in ymin : ymax ) {
        n[i,t] <- nonzeroweight[i]*exp(strata[i]+beta[i]*(t-fixedyear)+yeaffect[t,i]+ 0.5*sdnoise*sdnoise+
0.5*sdobs[i]*sdobs[i])
    } # end t-year loop
} # end i-strata loop

} # end model

```

Appendix 2 B.

BUGS language description of the trend comparison model

model for comparing trend estimates from two (indexed by e) different analyses, for species (indexed by s)
for the HB₃-ML comparison, e[1] are estimates from the HB₃ model and e[2] from the ML model
the trend comparison model is modified from the grassland bird model from Link and Barker 2010, pg 90, also described in Sauer and Link 2002, Ecology 83:1743-1751
in essence, this is a paired t-test style comparison, that accounts for the imprecision in each s*e trend estimate.

input data compiled in R, consist of nspecies (the number of species), as well as 3 matrices: varhat, betahat, and n, each of which has nspecies rows and 2 columns
varhat = estimates of the s*e variances of trends
betahat = estimates of the s*e trends
n = sample size of the s*e trends (for the HB₃-ML comparison, this is the number of routes used to generate the trend)

model{

for (e in 1:2) {

for(s in 1:nspecies) {

varhat[s,e] ~ dgamma(p[s,e],lam[s,e])
p[s,e] <- n[s,e] / 2
lam[s,e] <- p[s,e] * tau.betahat[s,e]
tau.betahat[s,e] ~ dgamma(0.001,0.001)
sd.betahat[s,e] <- 1/sqrt(tau.betahat[s,e])
betahat[s,e] ~ dnorm(beta[s,e],tau.betahat[s,e])
beta[s,e] ~ dnorm(0.0,1.0E-6)
pos[s,e] <- step(beta[s,e])
neg[s,e] <- 1-pos[s,e]

} # end of first s-species loop

numpos[e] <- sum(pos[,e])

numneg[e] <- sum(neg[,e])

} #end of e loop (indexing two models being compared)

for(s in 1:nspecies) {

dif[s] <- beta[s,1]-beta[s,2] # dif is a vector of the species-specific trend differences after accounting for the imprecision of each estimate's trend and the group (survey/monitoring program) structure

difvar[s] <- sd.betahat[s,1]-sd.betahat[s,2] # difvar is a vector of the species-specific differences in sd of the trends

posdifvar[s] <- step(difvar[s])

```
} # end of second s-species loop

p.HBimp <- (nspecies-sum(posdifvar[]))/nspecies

m.dif <- mean(dif[])
m.difvar <- mean(difvar[])

dif.numpos <- numpos[1]-numpos[2]

} # end of model
```